Name: **Huan Q. Bui**
Course: **8.309 - Classical Mechanics III**
Problem set: **#3**

1. Rotation Angle in the Euler Theorem

(a) Suppose $\vec{\xi}_1 = \vec{\xi}_2^* = (p + iq \quad m + in)^\top$ is normalized ($p, q, m, n \in \mathbb{R}$), then the combinations

$$\boxed{\vec{\xi}_a = \frac{1}{\sqrt{2}}\left[\vec{\xi}_1 + \vec{\xi}_2\right] \qquad \vec{\xi}_b = \frac{i}{\sqrt{2}}\left[\vec{\xi}_1 - \vec{\xi}_2\right]}$$

are what we want. First, $\xi_a \perp \xi_b$ by inspection and both are orthogonal to ξ_3 because they belong to the subspace spanned by $\vec{\xi}_1$ and $\vec{\xi}_2$ which is orthogonal to $\vec{\xi}_3$. Second, it is clear that both $\vec{\xi}_1$ and $\vec{\xi}_2$ are normalized. Finally, $\vec{\xi}_a$ and $\vec{\xi}_b$ are real because $\vec{\xi}_1 = \vec{\xi}_2^*$. We see that the components of $\vec{\xi}_a$ only involve the real part of $\vec{\xi}_1$ and $\vec{\xi}_2$, whereas the components of $\vec{\xi}_b$ only involve the imaginary parts of $\vec{\xi}_1$ and $\vec{\xi}_2$.

(b) The $(1 \quad 0 \quad 0)^\top$ vector in the $\{\vec{\xi}_1, \vec{\xi}_2, \vec{\xi}_3\}$ basis appears as $(1/\sqrt{2})(1 \quad -i \quad 0)$ in the $\{\vec{\xi}_a, \vec{\xi}_b, \vec{\xi}_3\}$ basis. Likewise, the $(0 \quad 1 \quad 0)^\top$ vector in the $\{\vec{\xi}_1, \vec{\xi}_2, \vec{\xi}_3\}$ basis appears as $(1/\sqrt{2})(1 \quad i \quad 0)$ in the $\{\vec{\xi}_a, \vec{\xi}_b, \vec{\xi}_3\}$ basis. Therefore, the matrix which connects the components in the former basis to the latter is

$$\boxed{W = \frac{1}{\sqrt{2}}\begin{pmatrix} 1 & 1 & 0 \\ -i & i & 0 \\ 0 & 0 & \sqrt{2} \end{pmatrix}}$$

By inspection, we see that all columns of W are mutually orthogonal with unit norm (with respect to the inner product for complex-valued vectors). So, W is unitary. Alternatively, we can explicitly calculate that $W^\dagger = W^{-1}$.

(c) With $\vec{u} = W\vec{s} = WX^\dagger\vec{r}$, we find

$$\vec{u}' = WX^\dagger\vec{r}' = WX^\dagger U\vec{r} = WX^\dagger UXW^\dagger\vec{u} =: \widetilde{U}\vec{u}.$$

So we have

$$\widetilde{U} = \frac{1}{2}\begin{pmatrix} 1 & 1 & 0 \\ -i & i & 0 \\ 0 & 0 & \sqrt{2} \end{pmatrix}\begin{pmatrix} e^{i\Phi} & & \\ & e^{-i\Phi} & \\ & & 1 \end{pmatrix}\begin{pmatrix} 1 & i & 0 \\ 1 & -i & 0 \\ 0 & 0 & \sqrt{2} \end{pmatrix} = \begin{pmatrix} \cos\Phi & -\sin\Phi & \\ \sin\Phi & \cos\Phi & \\ 0 & 0 & 1 \end{pmatrix}.$$

The matrix $\widetilde{U}$ has the form of a standard rotation matrix with rotation angle Φ, as desired. Note that defining the sign of $\vec{\xi}_b$ in Part (b) will impact whether the only $(-)$ sign in the matrix $\widetilde{U}$ goes on the (12) or (21) element. The answers in both cases are, of course, equivalent.

(d) Mathematica code:

```
(*Problem 1*)

In[7]:= W = (1/Sqrt[2])*{{1, 1, 0}, {-I, I, 0}, {0, 0, Sqrt[2]}};

In[8]:= ConjugateTranspose[W] - Inverse[W]

Out[8]= {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

In[10]:= ExpToTrig[
W . {{E^(I*\[Phi]), 0, 0}, {0, E^(-I*\[Phi]), 0}, {0, 0, 1}} . ConjugateTranspose[W]]

Out[10]= {{Cos[\[Phi]], -Sin[\[Phi]], 0}, {Sin[\[Phi]], Cos[\[Phi]], 0}, {0, 0, 1}}
```

2. Foucault Pendulum and the Coriolis Effect

(a) The Lagrangian is

$$\mathcal{L} = \frac{m}{2}\left[\vec{v} - \vec{\omega}\times(\vec{r} + R_e\hat{z})\right]^2 - V$$

where $\vec{v}$ is velocity in the rotating frame. According the definition of the spherical coordinates in this problem, we have

$$\vec{r} = \begin{pmatrix} x & y & z \end{pmatrix}^\top = \begin{pmatrix} l\sin\theta\sin\phi & l\sin\theta\cos\phi & -l\cos\theta \end{pmatrix}^\top$$

since θ is now defined as the angle from the $-\hat{z}$. In these coordinates, the potential energy is $V = -mgl\cos\theta$. The relevant vectors in this problem apart from $\vec{r}$ are

$$\vec{v} = \frac{d}{dt}\begin{pmatrix} x \\ y \\ z \end{pmatrix} = l\begin{pmatrix} \dot{\theta}\cos\theta\sin\phi + \sin\theta\dot{\phi}\cos\phi \\ \dot{\theta}\cos\theta\cos\phi - \sin\theta\dot{\phi}\sin\phi \\ -\dot{\theta}\sin\theta \end{pmatrix} \quad \text{and} \quad \vec{\omega} = \begin{pmatrix} 0 \\ \omega\cos\lambda \\ \omega\sin\lambda \end{pmatrix}$$

We can now plug everything into Mathematica to obtain the full Lagrangian. Under the approximation that $\omega R_e \to 0$ and $\omega^n \to 0$ for $n > 1$, we find the approximate Lagrangian to be

$$\mathcal{L} = \frac{1}{2}lm\left(2g\cos\theta + l\left(-2\omega\cos\lambda\dot{\theta}\cos 2\theta\sin\phi + \dot{\theta}^2 + \dot{\phi}\left(\sin^2\theta\left(\dot{\phi} - 2\omega\sin\lambda\right) - \omega\cos\lambda\sin 2\theta\cos\phi\right)\right)\right).$$

The equations of motion are:

$$\frac{d}{dt}\frac{\partial\mathcal{L}}{\partial\dot{\theta}} = \frac{\partial\mathcal{L}}{\partial\theta} \implies \boxed{\ddot{\theta} = \frac{-\sin\theta}{l}\left[g + l\cos\theta(2\omega\sin\lambda - \dot{\phi})\dot{\phi}\right]}$$

$$\frac{d}{dt}\frac{\partial\mathcal{L}}{\partial\dot{\phi}} = \frac{\partial\mathcal{L}}{\partial\phi} \implies \boxed{\ddot{\phi} = 2\cot\theta\dot{\theta}\left(\omega\sin\lambda - \dot{\phi}\right)}$$

−9

(b) Consider the small angle approximation where l is large, and θ is small. Then $\sin\theta \to \theta, \cos\theta \to 1$. Moreover, we will assume that locally the pendulum undergoes simple harmonic oscillation and therefore $\ddot{\theta}\theta \sim \theta^2 \to 0$. With these, the equations of motion become

$$\boxed{\ddot{\theta} = \frac{-\theta}{l}\left[g + l\dot{\phi}(2\omega\sin\lambda - \dot{\phi})\right]} \qquad \boxed{\ddot{\phi} = \frac{2\dot{\theta}}{\theta}\left[\omega\sin\lambda - \dot{\phi}\right]}$$

We notice that the first equation is a quadratic equation in $\dot{\phi}$. Solving it gives

$$\dot{\phi}_\pm = \omega\sin\lambda \pm \sqrt{\frac{g}{l}}.$$

Taking l to be arbitrarily large, we find the desired expression

$$\boxed{\dot{\phi} = \omega\sin\lambda}$$

(c) Mathematica code:

```
In[1]:= (*Problem 2*)

In[1]:= SphericalCoords = {x[t], y[t],
z[t]} -> {L*Sin[\[Theta][t]]*Sin[\[Phi][t]],
L*Sin[\[Theta][t]]*Cos[\[Phi][t]], -L*Cos[\[Theta][t]]};

In[3]:= SphericalDerivs = {D[x[t], t], D[y[t], t],
D[z[t], t]} -> {D[L*Sin[\[Theta][t]]*Sin[\[Phi][t]], t],
```

```
D[L*Sin[\[Theta][t]]*Cos[\[Phi][t]], t], D[L*Cos[\[Theta][t]], t]};

In[4]:= D[{x[t], y[t], z[t]}, t] /. SphericalDerivs // FullSimplify

Out[4]= {L (Cos[\[Theta][t]] Sin[\[Phi][t]] Derivative[1][\[Theta]][
t] + Cos[\[Phi][t]] Sin[\[Theta][t]] Derivative[1][\[Phi]][t]),
L (Cos[\[Theta][t]] Cos[\[Phi][t]] Derivative[1][\[Theta]][t] -
Sin[\[Theta][t]] Sin[\[Phi][t]] Derivative[1][\[Phi]][
t]), -L Sin[\[Theta][t]] Derivative[1][\[Theta]][t]}

In[5]:= (*Unit vectors in Cartesian*)

In[5]:= zHat = {0, 0, 1};

In[6]:= xHat = {1, 0, 0};

In[7]:= yHat = {0, 1, 0};

In[9]:= (*Relevant vectors*)

In[8]:= rVec = {L*Sin[\[Theta][t]]*Sin[\[Phi][t]],
L*Sin[\[Theta][t]]*Cos[\[Phi][t]], -L*Cos[\[Theta][t]]};

In[9]:= \[Omega]Vec = {0, \[Omega]*Cos[\[Lambda]], \[Omega]*
Sin[\[Lambda]]};

In[10]:= vVec = {D[x[t], t], D[y[t], t], D[z[t], t]} /.
SphericalDerivs;

In[11]:= wrRz = Cross[\[Omega]Vec, (rVec + RE*zHat)];

In[14]:= (*Potential*)

In[12]:= Pot = -m*g*L*Cos[\[Theta][t]] /. SphericalCoords;

In[16]:= (*Lagrangian*)

In[13]:= Lag = (m/2)*(Dot[vVec, vVec] + 2*Dot[vVec, wrRz] +
Dot[wrRz, wrRz]) - Pot // FullSimplify

Out[13]= 1/2 m (2 g L Cos[\[Theta][
t]] + \[Omega]^2 ((Cos[\[Lambda]] (RE - L Cos[\[Theta][t]]) -
L Cos[\[Phi][t]] Sin[\[Lambda]] Sin[\[Theta][t]])^2 +
L^2 Sin[\[Theta][t]]^2 Sin[\[Phi][t]]^2) +
2 L \[Omega] Cos[\[Lambda]] (RE Cos[\[Theta][t]] -
L Cos[2 \[Theta][t]]) Sin[\[Phi][t]] Derivative[1][\[Theta]][
t] + L^2 Derivative[1][\[Theta]][t]^2 -
2 L \[Omega] Sin[\[Theta][
t]] (Cos[\[Lambda]] (-RE + L Cos[\[Theta][t]]) Cos[\[Phi][t]] +
L Sin[\[Lambda]] Sin[\[Theta][t]]) Derivative[1][\[Phi]][t] +
L^2 Sin[\[Theta][t]]^2 Derivative[1][\[Phi]][t]^2)

In[14]:= Approx = {\[Omega]*RE -> 0,
RE*\[Omega] -> 0, \[Omega]^2 -> 0};

In[15]:= Lag = Lag /. Approx // FullSimplify

Out[15]= 1/2 L m (2 g Cos[\[Theta][t]] +
2 \[Omega] Cos[\[Lambda]] (RE Cos[\[Theta][t]] -
L Cos[2 \[Theta][t]]) Sin[\[Phi][t]] Derivative[1][\[Theta]][
t] + L Derivative[1][\[Theta]][t]^2 +
Sin[\[Theta][t]] Derivative[1][\[Phi]][
t] (2 \[Omega] Cos[\[Lambda]] (RE -
L Cos[\[Theta][t]]) Cos[\[Phi][t]] +
L Sin[\[Theta][t]] (-2 \[Omega] Sin[\[Lambda]] +
Derivative[1][\[Phi]][t])))

In[20]:= (*EOMs*)

In[21]:= (*\[Theta] equation*)

In[16]:= Eqn\[Theta] =
FullSimplify[D[D[Lag, \[Theta]'[t]], t] == D[Lag, \[Theta][t]]]

Out[16]= L m (Sin[\[Theta][t]] (g +
L Cos[\[Theta][t]] (2 \[Omega] Sin[\[Lambda]] -
Derivative[1][\[Phi]][t]) Derivative[1][\[Phi]][t]) +
L (\[Theta]^\[Prime]\[Prime])[t]) == 0

In[23]:= (*\[Phi] equation*)
```

```
In[17]:= Eqn\[Phi] =
FullSimplify[D[D[Lag, \[Phi]'[t]], t] == D[Lag, \[Phi][t]]]

Out[17]= L m (Sin[2 \[Theta][t]] Derivative[1][\[Theta]][
t] (\[Omega] Sin[\[Lambda]] - Derivative[1][\[Phi]][t]) -
Sin[\[Theta][t]]^2 (\[Phi]^\[Prime]\[Prime])[t]) == 0

In[18]:= Solve[Eqn\[Theta], \[Theta]''[t]] // FullSimplify

Out[18]= {{(\[Theta]^\[Prime]\[Prime])[t] -> -((
Sin[\[Theta][t]] (g +
L Cos[\[Theta][t]] (2 \[Omega] Sin[\[Lambda]] -
Derivative[1][\[Phi]][t]) Derivative[1][\[Phi]][t]))/L)}}

In[19]:= Solve[Eqn\[Phi], \[Phi]''[t]] // FullSimplify

Out[19]= {{(\[Phi]^\[Prime]\[Prime])[t] ->
2 Cot[\[Theta][t]] Derivative[1][\[Theta]][
t] (\[Omega] Sin[\[Lambda]] - Derivative[1][\[Phi]][t])}}

In[20]:= SmallAngle = {Sin[\[Theta][t]] -> \[Theta][t],
Cos[\[Theta][t]] -> 1,
Sin[2 \[Theta][t]] -> 2*\[Theta][t], \[Theta][t]*\[Theta]''[t] ->
0};

In[21]:= Eqn\[Theta] = Eqn\[Theta] /. SmallAngle /. Approx

Out[21]= L m (\[Theta][
t] (g + L (2 \[Omega] Sin[\[Lambda]] -
Derivative[1][\[Phi]][t]) Derivative[1][\[Phi]][t]) +
L (\[Theta]^\[Prime]\[Prime])[t]) == 0

In[22]:= Eqn\[Phi] = Eqn\[Phi] /. SmallAngle /. Approx

Out[22]= L m (2 \[Theta][t] Derivative[1][\[Theta]][
t] (\[Omega] Sin[\[Lambda]] -
Derivative[1][\[Phi]][t]) - \[Theta][
t]^2 (\[Phi]^\[Prime]\[Prime])[t]) == 0

In[23]:= Solve[Eqn\[Theta], \[Theta]''[t]] // FullSimplify

Out[23]= {{(\[Theta]^\[Prime]\[Prime])[
t] -> -((\[Theta][
t] (g + L (2 \[Omega] Sin[\[Lambda]] -
Derivative[1][\[Phi]][t]) Derivative[1][\[Phi]][t]))/L)}}

In[25]:= Solve[Eqn\[Phi], \[Phi]''[t]] // FullSimplify

Out[25]= {{(\[Phi]^\[Prime]\[Prime])[t] -> (
2 Derivative[1][\[Theta]][
t] (\[Omega] Sin[\[Lambda]] -
Derivative[1][\[Phi]][t]))/\[Theta][t]}}

In[28]:= (*Equation for \[Phi]'[t]. When L -> \[Infinity] then we get \
desired result*)

In[26]:= FullSimplify[
Solve[Eqn\[Theta], \[Phi]'[t]] /. SmallAngle /. Approx //
Expand, {L > 0, \[Theta][t] > 0, g > 0}]

Out[26]= {{Derivative[1][\[Phi]][
t] -> -(g/Sqrt[g L]) + \[Omega] Sin[\[Lambda]]}, {Derivative[
1][\[Phi]][t] -> Sqrt[g/L] + \[Omega] Sin[\[Lambda]]}}
```

3. Angular Velocity with Euler Angles

(a) Since $\vec{\omega}_\phi$ is parallel to the space z-axis, it has the form $\vec{\omega}_\phi = (0 \quad 0 \quad \dot{\phi})^\top$ in the space-basis. To go from the space basis to the body basis we simply apply the full orthogonal transformation $A = BCD$. The result is simply the third column of A, mutiplied by $\dot{\phi}$:

$$\begin{pmatrix} (\omega_\phi)_{x'} \\ (\omega_\phi)_{y'} \\ (\omega_\phi)_{z'} \end{pmatrix} = \begin{pmatrix} \dot{\phi}\sin\theta\sin\psi \\ \dot{\phi}\sin\theta\cos\psi \\ \dot{\phi}\cos\theta \end{pmatrix}.$$

Next, since $\vec{\omega}_\theta$ is in line with the ξ' axis (which is the space x-axis after the orthogonal transformations D and C), we only need to apply the final orthogonal transformation B to $CD\vec{\omega}_\theta = (\dot{\theta} \quad 0 \quad 0)^\top$. This give the first column of B, multiplied by $\dot{\theta}$:

$$\begin{pmatrix} (\omega_\theta)_{x'} \\ (\omega_\theta)_{y'} \\ (\omega_\theta)_{z'} \end{pmatrix} = \begin{pmatrix} \dot{\theta}\cos\psi \\ -\dot{\theta}\sin\psi \\ 0 \end{pmatrix}.$$

Finally, $\vec{\omega}_\psi$ lies along the z'-axis, so no transformation is necessary.

$$\begin{pmatrix} (\omega_\psi)_{x'} \\ (\omega_\psi)_{y'} \\ (\omega_\psi)_{z'} \end{pmatrix} = \begin{pmatrix} 0 \\ 0 \\ \dot{\psi} \end{pmatrix}$$

Adding everything together, we find

$$\begin{aligned} \omega_{x'} &= \dot{\phi}\sin\theta\sin\psi + \dot{\theta}\cos\psi \\ \omega_{y'} &= \dot{\phi}\sin\theta\cos\psi - \dot{\theta}\sin\psi \\ \omega_{z'} &= \dot{\phi}\cos\theta + \dot{\psi} \end{aligned}$$

(b) From the previous part, we have $\vec{\omega}' = (\omega_{x'} \quad \omega_{y'} \quad \omega_{z'})$. To transform these into $\vec{\omega} = (\omega_x \quad \omega_y \quad \omega_z)$ we simply apply A^{-1}, since $\vec{\omega}' = A\vec{\omega}$. To reduce our chance of making a mistake, we may write $A^{-1} = (BCD)^{-1} = D^{-1}C^{-1}B^{-1}$ where

$$D = \begin{pmatrix} \cos\phi & \sin\phi & 0 \\ -\sin\phi & \cos\phi & 0 \\ 0 & 0 & 1 \end{pmatrix} \qquad C = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos\theta & \sin\theta \\ 0 & -\sin\theta & \cos\theta \end{pmatrix} \qquad B = \begin{pmatrix} \cos\psi & \sin\psi & 0 \\ -\sin\psi & \cos\psi & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Plugging everything we find that

$$\vec{\omega} = \begin{pmatrix} \omega_x \\ \omega_y \\ \omega_z \end{pmatrix} = D^{-1}C^{-1}B^{-1}\begin{pmatrix} \omega_{x'} \\ \omega_{y'} \\ \omega_{z'} \end{pmatrix} = \begin{pmatrix} \dot{\theta}\cos\phi + \dot{\psi}\sin\theta\sin\phi \\ \dot{\theta}\sin\phi - \dot{\psi}\sin\theta\cos\phi \\ \dot{\psi}\cos\theta + \dot{\phi} \end{pmatrix},$$

as desired.

(c) Working in the body axes, the moment of inertia tensor is diagonal $\hat{I} = \mathrm{diag}(I_{x'}, I_{y'}, I_{z'})$. The kinetic energy is then simply

$$\begin{aligned} T &= \frac{1}{2}\vec{\omega}'^\top \hat{I}\vec{\omega} \\ &= \frac{1}{2}\left[I_{x'}\omega_{x'}^2 + I_{y'}\omega_{y'}^2 + I_{z'}\omega_{z'}^2\right] \\ &= \frac{1}{2}\left[I_{x'}(\cos\psi\dot{\theta} + \sin\theta\sin\psi\dot{\phi})^2 + I_{y'}(\sin\psi\dot{\theta} - \cos\psi\sin\theta\dot{\phi})^2 + I_{z'}(\cos\theta\dot{\phi} + \dot{\psi})^2\right]. \end{aligned}$$

Using the generalied coordinate ψ, the Euler-Lagrange equation in the problem becomes

$$I_{z'}\left[-\sin\theta\dot{\theta}\dot{\phi} + \cos\theta\ddot{\phi} + \ddot{\psi}\right] - (I_{x'} - I_{y'})\left[-\cos\psi\sin\psi\dot{\theta}^2 + \cos 2\psi\sin\theta\dot{\theta}\dot{\phi} + \cos\psi\sin^2\theta\sin\psi\dot{\phi}^2\right] = Q_{z'}$$

where $q_j = q_{z'} = \psi$, and $Q_j = Q_{z'}$. Now notice that

$$\dot{\omega}_{z'} = -\sin\theta\dot{\theta}\dot{\phi} + \cos\theta\ddot{\phi} + \ddot{\psi} \quad \& \quad \omega_{x'}\omega_{y'} = -\cos\psi\sin\psi\dot{\theta}^2 + \cos 2\psi\sin\theta\dot{\theta}\dot{\phi} + \cos\psi\sin^2\theta\sin\psi\dot{\phi}^2.$$

Thus, we find

$$\boxed{I_{z'}\omega_{z'} - \omega_{x'}\omega_{y'}(I_{x'} - I_{y'}) = Q_{z'}}$$

which is what we would expect from deriving the equations of motion from Newtonian mechanics.

(d) Mathematica code:

```
(*Problem 3*)

In[2]:= d = {{Cos[\[Phi]], Sin[\[Phi]], 0}, {-Sin[\[Phi]],
Cos[\[Phi]], 0}, {0, 0, 1}};

In[3]:= c = {{1, 0, 0}, {0, Cos[\[Theta]],
Sin[\[Theta]]}, {0, -Sin[\[Theta]], Cos[\[Theta]]}};

In[5]:= b = {{Cos[\[Psi]], Sin[\[Psi]], 0}, {-Sin[\[Psi]],
Cos[\[Psi]], 0}, {0, 0, 1}};

In[12]:= \[Omega]Prime = {\[Phi]'*Sin[\[Theta]]*
Sin[\[Psi]] + \[Theta]'*Cos[\[Psi]], \[Phi]'*Sin[\[Theta]]*
Cos[\[Psi]] - \[Theta]'*Sin[\[Psi]], \[Phi]'*
Cos[\[Theta]] + \[Psi]'};

In[13]:= \[Omega] =
Inverse[d] . Inverse[c] . Inverse[b] . \[Omega]Prime // FullSimplify

Out[13]= {Cos[\[Phi]] Derivative[1][\[Theta]] +
Sin[\[Theta]] Sin[\[Phi]] Derivative[1][\[Psi]],
Sin[\[Phi]] Derivative[1][\[Theta]] -
Cos[\[Phi]] Sin[\[Theta]] Derivative[1][\[Psi]],
Derivative[1][\[Phi]] + Cos[\[Theta]] Derivative[1][\[Psi]]}

In[31]:= \[Omega]1 = \[Phi]'[t]*Sin[\[Theta][t]]*
Sin[\[Psi][t]] + \[Theta]'[t]*Cos[\[Psi][t]];

In[40]:= \[Omega]2 = \[Phi]'[t]*Sin[\[Theta][t]]*
Cos[\[Psi][t]] - \[Theta]'[t]*Sin[\[Psi][t]];

In[41]:= \[Omega]3 = \[Phi]'[t]*Cos[\[Theta][t]] + \[Psi]'[t];

In[51]:= T = (1/2)*(I1*\[Omega]1^2 + I2*\[Omega]2^2 +
I3*\[Omega]3^2) // Simplify

Out[51]= 1/2 (I2 (Sin[\[Psi][t]] Derivative[1][\[Theta]][t] -
Cos[\[Psi][t]] Sin[\[Theta][t]] Derivative[1][\[Phi]][t])^2 +
I1 (Cos[\[Psi][t]] Derivative[1][\[Theta]][t] +
Sin[\[Theta][t]] Sin[\[Psi][t]] Derivative[1][\[Phi]][t])^2 +
I3 (Cos[\[Theta][t]] Derivative[1][\[Phi]][t] +
Derivative[1][\[Psi]][t])^2)

In[43]:= (*LHS of EOM using \[Psi]*)

In[52]:= D[D[T, \[Psi]'[t]], t] - D[T, \[Psi][t]] // FullSimplify

Out[52]= (I1 - I2) Cos[\[Psi][t]] Sin[\[Psi][t]] Derivative[
1][\[Theta]][
t]^2 - (I3 + (I1 - I2) Cos[2 \[Psi][t]]) Sin[\[Theta][
t]] Derivative[1][\[Theta]][t] Derivative[1][\[Phi]][
t] + (-I1 + I2) Cos[\[Psi][t]] Sin[\[Theta][t]]^2 Sin[\[Psi][
t]] Derivative[1][\[Phi]][t]^2 +
I3 (Cos[\[Theta][t]] (\[Phi]^\[Prime]\[Prime])[
t] + (\[Psi]^\[Prime]\[Prime])[t])

(*Sanity check*)

In[45]:= \[Omega]1*\[Omega]2 // FullSimplify

Out[45]= (-Sin[\[Psi][t]] Derivative[1][\[Theta]][t] +
Cos[\[Psi][t]] Sin[\[Theta][t]] Derivative[1][\[Phi]][
t]) (Cos[\[Psi][t]] Derivative[1][\[Theta]][t] +
Sin[\[Theta][t]] Sin[\[Psi][t]] Derivative[1][\[Phi]][t])

In[47]:= D[\[Omega]3, t] // FullSimplify

Out[47]= -Sin[\[Theta][t]] Derivative[1][\[Theta]][t] Derivative[
1][\[Phi]][t] +
Cos[\[Theta][t]] (\[Phi]^\[Prime]\[Prime])[
t] + (\[Psi]^\[Prime]\[Prime])[t]

In[56]:= (*Simplifications to find \[Omega]3*)

In[60]:= D[\[Omega]3,
t] - (- Sin[\[Theta][t]] Derivative[1][\[Theta]][t] Derivative[
1][\[Phi]][
t] + (Cos[\[Theta][t]] (\[Phi]^\[Prime]\[Prime])[
t] + (\[Psi]^\[Prime]\[Prime])[t])) // FullSimplify
```

```
Out[60]= 0

(*Simplifications to find \[Omega]1*\[Omega]2*)

In[66]:= \[Omega]1*\[Omega]2 - (-Cos[\[Psi][t]] Sin[\[Psi][
t]] Derivative[1][\[Theta]][t]^2 +
Cos[2 \[Psi][t]]*
Sin[\[Theta][t]] Derivative[1][\[Theta]][t] Derivative[
1][\[Phi]][t] +
Cos[\[Psi][t]] Sin[\[Theta][t]]^2 Sin[\[Psi][t]] Derivative[
1][\[Phi]][t]^2) // FullSimplify

Out[66]= 0
```

4. Point Mass on a Disk

(a) By inspection, we can pick out the principal axes about the disk's center as the usual $x-, y-, z-$axes where the origin is at the disk's center and z points out of the page. By symmetry, we only need to calculate the moment of inertia for rotations about z and rotations about x.

$$I_{zz,\text{disk}} = \int_A \frac{M}{\pi R^2}(r^2 - \cancel{z^2})\, dA = \frac{M}{\pi R^2}\int_{r=0}^{R}\int_0^{2\pi} r^3\, d\theta\, dr = \frac{1}{2}MR^2.$$

$$I_{xx,\text{disk}} = I_{yy,\text{disk}} = \int_A \frac{M}{\pi R^2}(r^2 - xx)\, dA = \frac{M}{\pi R^2}\int_{r=0}^{R}\int_0^{2\pi} r^3(1-\sin^2\theta)\, d\theta\, dr = \frac{1}{4}MR^2.$$

So, the moment of inertia tensor of the disk about its center is

$$\boxed{\hat{I}_{\text{disk, disk center}} = \text{diag}\left[\frac{1}{4}MR^2, \frac{1}{4}MR^2, \frac{1}{2}MR^2\right]}$$

Now we wish to compute the moment of inertia of the disk about point A. To this end, we use the parallel axis theorem.

$$\hat{I}^A_{ab} = M(\delta_{ab}\mathbf{R}^2 - \mathbf{R}_a\mathbf{R}_b) + \hat{I}^{\text{center}}_{ab}$$

where $\mathbf{R} = (0, R, 0)^\top$ denotes the translation vector from Q to the center of the disk. It is clear that there is no change to $\hat{I}$ whenever $a \neq b$. The only possible changes are along in xx, yy, zz:

$$\hat{I}^A_{xx} = MR^2 + I^{\text{center}}_{xx} = \frac{5}{4}MR^2$$
$$\hat{I}^A_{yy} = M(R^2 - R^2) + I^{\text{center}}_{yy} = \frac{1}{4}MR^2$$
$$\hat{I}^A_{zz} = MR^2 + I^{\text{center}}_{zz} = \frac{3}{2}MR^2,$$

Finally, we calculate the moment of inertia tensor of the mass m, which is attached on the disk, about the point A. The distance from the mass to point A is $r = R\sqrt{2}$. Based on the geometry of the problem we also know that $y = x = R$ and $z = 0$. We shall proceed:

$$\hat{I}^A_m = \frac{3}{8}M\begin{pmatrix} r^2 - x^2 & -xy & -xz \\ -xy & r^2 - y^2 & -yz \\ -xz & -yz & r^2 - z^2 \end{pmatrix} = \frac{3}{8}M\begin{pmatrix} 2R^2 - R^2 & -R^2 & 0 \\ -R^2 & 2R^2 - R^2 & 0 \\ 0 & 0 & 2R^2 \end{pmatrix} = \frac{3}{8}MR^2\begin{pmatrix} 1 & -1 & 0 \\ -1 & 1 & 0 \\ 0 & 0 & 2 \end{pmatrix}.$$

With this, the combined moment of inertia of the system about point A is

$$\hat{I}^A_{\text{system}} = MR^2\begin{pmatrix} 3/8 + 5/4 & -3/8 & 0 \\ -3/8 & 3/8 + 1/4 & 0 \\ 0 & 0 & 6/8 + 3/2 \end{pmatrix} = \boxed{\frac{MR^2}{8}\begin{pmatrix} 13 & -3 & 0 \\ -3 & 5 & 0 \\ 0 & 0 & 18 \end{pmatrix}}$$

(b) To find the principal axes and principal moments of inertia, we simply have to diagonalize $\hat{I}^A_{\text{system}}$ found in the previous part. We may use Mathematica to do this. The principal moment of inertia tensor is

$$\boxed{\hat{I}^{\text{principal}}_{\text{system}} = MR^2 \text{diag}\left[\frac{9}{4}, \frac{7}{4}, \frac{1}{2}\right]}$$

where the corresponding normalized principal axes are

$$\boxed{\begin{pmatrix}0\\0\\1\end{pmatrix}, \quad \frac{1}{\sqrt{10}}\begin{pmatrix}-3\\1\\0\end{pmatrix}, \quad \frac{1}{\sqrt{10}}\begin{pmatrix}1\\3\\0\end{pmatrix}}$$

To do all this by hand one will need to write out the characteristic equation for $\hat{I}^A_{\text{system}}$ and solve for the eigenvalues, which are the roots of that equation. Then, for each eigenvalue λ, one will find the kernel of the matrix $\hat{I}^A_{\text{system}} - \lambda\mathbb{I}$ and then find an orthonormal basis for each subspace. In principle, the various subspaces corresponding to distinct eigenvalues will be orthogonal, so concatenating the basis vectors will gives us an orthonormal basis aka a set of principal axes.

(c) In the coordinate system centered at A, the angular velocity vector of the disk plus mass system is simply $\vec{\omega}_A = (0, \omega, 0)^\top$. The angular momentum about point A is then

$$\vec{L}^A = \hat{I}^A\vec{\omega} = \frac{\omega MR^2}{8}\begin{pmatrix}-3\\5\\0\end{pmatrix}$$

Here, $\vec{L}^A$ is the angular momentum in the body frame which is attached to the system. To calculate the angular momentum of the system in the lab frame, we must transform out of the body frame via a time-dependent rotation matrix about y. Let the lab frame be (x_l, y_l, z_l). The transformation matrix

$$\begin{pmatrix}x_l\\y_l\\z_l\end{pmatrix} = \begin{pmatrix}\cos\omega t & 0 & \sin\omega t\\ 0 & 1 & 0\\ -\sin\omega t & 0 & \cos\omega t\end{pmatrix}\begin{pmatrix}x\\y\\z\end{pmatrix}.$$

With this, the angular momentum vector in the lab frame is given by

$$\vec{L}_{\text{lab}} = \frac{\omega MR^2}{8}\begin{pmatrix}\cos\omega t & 0 & \sin\omega t\\ 0 & 1 & 0\\ -\sin\omega t & 0 & \cos\omega t\end{pmatrix}\begin{pmatrix}-3\\5\\0\end{pmatrix} = \boxed{\frac{\omega MR^2}{8}\begin{pmatrix}-3\cos\omega t\\5\\3\sin\omega t\end{pmatrix}}$$

(d) Mathematica code:

```
In[68]:= (*Problem 4*)

In[69]:= Integrate[(M/(Pi*R^2))*r^3, {r, 0, R}, {\[Theta], 0, 2*Pi}]

Out[69]= (M R^2)/2

In[70]:= Integrate[(M/(Pi*R^2))*r^3 (1 - Sin[\[Theta]]^2), {r, 0,
R}, {\[Theta], 0, 2*Pi}]

Out[70]= (M R^2)/4

In[95]:= Ihat =
M*R^2*{{3/8 + 5/4, -3/8, 0}, {-3/8, 3/8 + 1/4, 0}, {0, 0,
2*3/8 + 3/2}} // FullSimplify

Out[95]= {{(13 M R^2)/8, -((3 M R^2)/8), 0}, {-((3 M R^2)/8), (
5 M R^2)/8, 0}, {0, 0, (9 M R^2)/4}}

In[96]:= Eigensystem[Ihat] // Simplify

Out[96]= {{(9 M R^2)/4, (7 M R^2)/4, (M R^2)/
2}, {{0, 0, 1}, {-3, 1, 0}, {1, 3, 0}}}
```

5. A Rolling Cone (Goldstein Ch.5 #17)

(a) The volume of the cone is well known: $V = \pi R^2 h/3$. To find the center of mass of the cone, we use the following definition

$$\vec{r}_{\text{CM}} = \frac{1}{\int \rho\, dV} \int \vec{r}\, dm = \frac{1}{V} \int \vec{r}\, dV.$$

By symmetry, the center of mass must lie on the axis which goes through the tip and the center of the base. So, choosing our coordinates such that the origin is at the tip and the z-axis goes through the center of the base, we find that it suffices to find where the center of mass lies on the z-axis. To do this, we simply compute:

$$z_{\text{CM}} = \frac{1}{V} \int_0^h z \underbrace{\left(\pi z^2 \frac{R^2}{h^2}\right) dz}_{dV} = \frac{3h}{4}.$$

So, the center of mass is on axis which goes through the tip and the center of the base of the cone, at a distance of $3h/4$ from the tip.

Now pick a new set of axes so that the y'-axis goes through the tip and the center of mass as required by the problem. The moment of inertia tensor is obtained using the following formula

$$I_{ab} = \int \rho(\vec{r}) \left[\vec{r}^2 \delta_{ab} - \vec{r}_a \vec{r}_b\right] dV.$$

where $\vec{r}_a$ denotes the components of the position vector $\vec{r}$. By symmetry, we may use cylindrical coordinates where

$$(x', y', z') = (r\cos\phi, y', r\sin\phi).$$

To make things look a bit more compact, let the mass of the cone be $M = \rho V$. The various components of $\hat{I}$ are

$$\begin{aligned}
I_{x'x'} &= \frac{M}{V} \int_0^h \int_0^{Ry'/h} \int_0^{2\pi} (r^2 + y'^2 - r^2\cos^2\phi) r\, d\phi\, dr\, dy' = \frac{3}{20} M(4h^2 + R^2) \\
I_{y'y'} &= \frac{M}{V} \int_0^h \int_0^{Ry'/h} \int_0^{2\pi} (r^2 + y'^2 - y'^2) r\, d\phi\, dr\, dy' = \frac{3}{10} MR^2 \\
I_{z'z'} &= \frac{M}{V} \int_0^h \int_0^{Ry'/h} \int_0^{2\pi} (r^2 + y'^2 - r^2\sin^2\phi) r\, d\phi\, dr\, dy' = \frac{3}{20} M(4h^2 + R^2) \\
I_{x'y'} &= \frac{M}{V} \int_0^h \int_0^{Ry'/h} \int_0^{2\pi} (-ry'\cos\phi) r\, d\phi\, dr\, dy' = 0 \\
I_{x'z'} &= \frac{M}{V} \int_0^h \int_0^{Ry'/h} \int_0^{2\pi} (-r^2\cos\phi\sin\phi) r\, d\phi\, dr\, dy' = 0 \\
I_{y'z'} &= \frac{M}{V} \int_0^h \int_0^{Ry'/h} \int_0^{2\pi} (-ry'\sin\phi) r\, d\phi\, dr\, dy' = 0.
\end{aligned}$$

So,

$$\boxed{\hat{I} = \text{diag}\left[\frac{3}{20} M(4h^2 + R^2), \frac{3}{10} MR^2, \frac{3}{20} M(4h^2 + R^2)\right]}$$

(b) Now let us move the axes to the center of mass. The displacement vector is thus $\vec{k} = (0, 3h/4, 0)$, and we do not introduce any off-diagonal terms to the new moment of inertia tensor. By the parallel axes theorem, the new moment of inertia tensor is given by

$$\begin{aligned}
\hat{I}^{\text{CM}}_{ab} &= \hat{I}_{ab} - M\left(\delta_{ab}\vec{k}^2 - \vec{k}_a\vec{k}_b\right) \\
&= \text{diag}\left[\frac{3}{20}M(4h^2+R^2) - \frac{9Mh^2}{16}, \frac{3}{10}MR^2, \frac{3}{20}M(4h^2+R^2) - \frac{9Mh^2}{16}\right] \\
&= \boxed{\text{diag}\left[\frac{3}{80}M(h^2+4R^2), \frac{3}{10}MR^2, \frac{3}{80}M(h^2+4R^2)\right]}
\end{aligned}$$

(c) The new set of axes (x, y, z) is simply the old set of axes (x', y', z') rotated clockwise by the angle α about the x axis. Therefore,

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} x \\ y'\cos\alpha + z'\sin\alpha \\ -y'\sin\alpha + z'\cos\alpha \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos\alpha & \sin\alpha \\ 0 & -\sin\alpha & \cos\alpha \end{pmatrix}\begin{pmatrix} x' \\ y' \\ z' \end{pmatrix} =: R_x\begin{pmatrix} x' \\ y' \\ z' \end{pmatrix}$$

The new moment of inertia is thus

$$\boxed{\hat{I}_{\text{unprimed}} = R_x\hat{I}R_x^{\top} = \frac{3M}{80}\begin{pmatrix} h^2+4R^2 & 0 & 0 \\ 0 & 2R^2(3+\cos 2\alpha) + h^2\sin^2\alpha & (h^2-4R^2)\cos\alpha\sin\alpha \\ 0 & (h^2-4R^2)\cos\alpha\sin\alpha & (h^2+4R^2)\cos^2\alpha + 8R^2\sin^2\alpha \end{pmatrix}}$$

(d) Let us use Part (b) to find the kinetic energy of the rolling cone, then use Part (c) to confirm our finding. In view of Part (b), we may split the kinetic energy T into the translational part of the center of mass and rotation part about the center of mass.

$$T = T^{\text{CM}}_{\text{trans}} + T^{\text{CM}}_{\text{rot}}.$$

The center of mass moves in a circular path at constant speed. Let ω_z denote the angular velocity of the center of mass (which points in the z direction). Then the translational kinetic energy is given by

$$T^{\text{CM}}_{\text{trans}} = \frac{1}{2}M\left(\frac{3h}{4}\cos\alpha\right)^2\omega_z^2.$$

Here, the speed at which the center of mass travels is

$$v = \frac{3h}{4}\cos\alpha\omega_z.$$

As the cone rolls without slipping, it is instantaneously rotating about the axis which coincides with the line of contact with angular velocity $\vec{\omega}$, which points in the y direction in view of Part (c). In the (x', y', z') body axes, this angular velocity can be written as

$$(0, \omega\cos\alpha, \omega\sin\alpha)$$

where we have let ω denote the magnitude of $\vec{\omega}$. From here, we have two equivalent expressions for the speed of the center of mass

$$v = \frac{3h}{4}\cos\alpha\omega_z = \frac{3h}{4}\sin\alpha\omega.$$

which implies that

$$\omega = \omega_z\cot\alpha.$$

We can now write down the kinetic energy of the rolling cone. With $R/h = \tan\alpha$, we have

$$
\begin{aligned}
T &= T_{\text{trans}}^{\text{CM}} + \frac{1}{2}\left[I_{z'z'}\omega_{z'}^2 + I_{y'y'}\omega_{y'}^2\right] \\
&= \frac{1}{2}M\left(\frac{3h}{4}\cos\alpha\right)^2\omega_z^2 + \frac{1}{2}\left[\frac{3MR^2}{10}(\omega_z\cos\alpha\cot\alpha)^2 + \frac{3M}{80}\left(h^2+4R^2\right)(\omega_z\sin\alpha\cot\alpha)^2\right] \\
&= \boxed{\frac{3h^2M\omega_z^2}{80}(7+5\cos 2\alpha)} \\
&= \boxed{\frac{3h^2M\omega_z^2}{40}\left(1+5\cos^2\alpha\right)}
\end{aligned}
$$

I have left my answer in different trigonometric forms so it is easier for the grader.

As a sanity check, let us use the result of Part (c) to get the same answer. Instead of decomposing $\vec{\omega}$ into components in the (x', y', z') basis, we can directly use $\hat{I}_{\text{unprimed}}$. In the (x, y, z) basis, $\vec{\omega}$ is simply $(0, \omega, 0)^\top = (0, \omega_z\cot\alpha, 0)^\top$.

$$
\begin{aligned}
T &= T_{\text{trans}}^{\text{CM}} + \frac{1}{2}\vec{\omega}^\top\hat{I}_{\text{unprimed}}\vec{\omega} \\
&= T_{\text{trans}}^{\text{CM}} + \frac{1}{2}\hat{I}_{\text{unprimed},yy}\omega_z^2\cot^2\alpha \\
&= \frac{1}{2}M\left(\frac{3h}{4}\cos\alpha\right)^2\omega_z^2 + \frac{1}{2}\frac{3M}{80}\left[2R^2(3+\cos 2\alpha) + h^2\sin^2\alpha\right]\omega_z^2\cot^2\alpha \\
&= \frac{3h^2M\omega_z^2}{80}(7+5\cos 2\alpha)\,,
\end{aligned}
$$

as expected (where we have once again used $R/h = \tan\alpha$). So, all is good.

(e) Mathematica code:

```
In[69]:= (*Problem 5*)

In[70]:= (*Volume of a cone*)

In[21]:= Vol = (1/3)*h*Pi*R^2;

In[71]:= (*COM position on z*)

In[72]:= (1/M) (M/Vol)*Integrate[Pi*z*z^2*(R/h)^2, {z, 0, h}]

Out[72]= (3 h)/4

In[73]:= (*Moment of inertia tensor elements*)

In[37]:= Ixx = (M/Vol)*
Integrate[(r^2 + y^2 - r^2*Cos[\[Theta]]^2)*r, {y, 0, h}, {r, 0,
y*R/h}, {\[Theta], 0, 2*Pi}] // FullSimplify

Out[37]= 3/20 M (4 h^2 + R^2)

In[39]:= Izz = (M/Vol)*
Integrate[(r^2 + y^2 - r^2*Sin[\[Theta]]^2)*r, {\[Theta], 0,
2*Pi}, {y, 0, h}, {r, 0, y*R/h}] // FullSimplify

Out[39]= 3/20 M (4 h^2 + R^2)

In[27]:= Iyy = (M/Vol)*
Integrate[(r^2 + y^2 - y^2)*r, {\[Theta], 0, 2*Pi}, {y, 0, h}, {r,
0, y*R/h}] // FullSimplify

Out[27]= (3 M R^2)/10

In[28]:= Ixy = (M/Vol)*
Integrate[(-r*Cos[\[Theta]]*y)*r, {\[Theta], 0, 2*Pi}, {y, 0,
h}, {r, 0, y*R/h}] // FullSimplify
```

```
Out[28]= 0

In[29]:= Ixz = (M/Vol)*
Integrate[(-r^2*Cos[\[Theta]]*Sin[\[Theta]])*r, {\[Theta], 0,
2*Pi}, {y, 0, h}, {r, 0, y*R/h}] // FullSimplify

Out[29]= 0

In[30]:= Iyz = (M/Vol)*
Integrate[(-y*r*Sin[\[Theta]])*r, {\[Theta], 0, 2*Pi}, {y, 0,
h}, {r, 0, y*R/h}] // FullSimplify

Out[30]= 0

In[42]:= Ixx - M*9*h^2/16 // Simplify

Out[42]= 3/80 M (h^2 + 4 R^2)

In[74]:= (*Rotation matrix in X*)

In[75]:= Rx = {{1, 0, 0}, {0, Cos[\[Alpha]],
Sin[\[Alpha]]}, {0, -Sin[\[Alpha]], Cos[\[Alpha]]}};

In[79]:= (*Moment of Inertia Tensor*)

In[77]:= Ihat = {{(3 M/80)*(h^2 + 4 R^2), 0, 0}, {0, (3/10) M*R^2,
0}, {0, 0, (3 M/80)*(h^2 + 4 R^2)}};

In[80]:= (*Transformed Moment of Inertia Tensor*)

In[47]:= Rx . Ihat . Transpose[Rx] // FullSimplify

Out[47]= {{3/80 M (h^2 + 4 R^2), 0, 0}, {0,
3/80 M (2 R^2 (3 + Cos[2 \[Alpha]]) + h^2 Sin[\[Alpha]]^2),
3/80 M (h^2 - 4 R^2) Cos[\[Alpha]] Sin[\[Alpha]]}, {0,
3/80 M (h^2 - 4 R^2) Cos[\[Alpha]] Sin[\[Alpha]],
3/80 M ((h^2 + 4 R^2) Cos[\[Alpha]]^2 + 8 R^2 Sin[\[Alpha]]^2)}}

In[64]:= (*Kinetic Energy*)

In[81]:= T = (1/
2)*((3 M/80)*(h^2 +
4*R^2)*(\[Omega]z*Sin[\[Alpha]]*Cot[\[Alpha]])^2 + (3 M/10)*
R^2*(\[Omega]z*Cos[\[Alpha]]*Cot[\[Alpha]])^2) + (1/2)*
M*(3*h*Cos[\[Alpha]]/4)^ 2*\[Omega]z^2 /. {R ->
h*Tan[\[Alpha]]} // FullSimplify

Out[81]= 3/80 h^2 M \[Omega]z^2 (7 + 5 Cos[2 \[Alpha]])

In[82]:= (*Sanity check: using transformed moment of inertia tensor*)

In[84]:= TT = (1/2)*
M*(3*h*Cos[\[Alpha]]/4)^
2*\[Omega]z^2 + (1/2)*(\[Omega]z*Cot[\[Alpha]])^2*(3/
80 M (2 R^2 (3 + Cos[2 \[Alpha]]) +
h^2 Sin[\[Alpha]]^2)) /. {R -> h*Tan[\[Alpha]]} //
FullSimplify

Out[84]= 3/80 h^2 M \[Omega]z^2 (7 + 5 Cos[2 \[Alpha]])
```