Name: **Huan Q. Bui**
Course: **8.309 - Classical Mechanics III**
Problem set: **#1**

1. Two Particles in a Gravitational Field

(a) In the center of mass (COM) frame, the Lagrangian is given by

$$\begin{aligned}\mathcal{L} &= T - U \\ &= T - U_{\text{attraction}} - U_g \\ &= \left[\frac{1}{2}m_1\dot{\vec{r}}_1^2 + \frac{1}{2}m_2\dot{\vec{r}}_2^2\right] - \left[-\frac{Gm_1m_2}{|\vec{r}|} - g(m_1x_1 + m_2x_2)\right] \\ &= \frac{1}{2}(m_1+m_2)\dot{\vec{R}}_{\text{COM}}^2 + \frac{1}{2}\frac{m_1m_2}{m_1+m_2}\dot{\vec{r}}^2 + \frac{Gm_1m_2}{r} + g(m_1+m_2)X_{\text{COM}},\end{aligned}$$

where X_{COM} is the x-component of $\vec{R}_{\text{COM}}$. To obtain this Lagrangian, we have solved for $\vec{r}_1$ and $\vec{r}_2$ in terms of $\vec{R}_{\text{COM}}$ and $\vec{r}$ from the following definitions:

$$\begin{cases}\vec{R}_{\text{COM}} = (m_1\vec{r}_1 + m_2\vec{r}_2)/(m_1+m_2) \\ \vec{r} = \vec{r}_2 - \vec{r}_1\end{cases} \implies \begin{cases}\vec{r}_1 = \vec{R}_{\text{COM}} - m_2\vec{r}/(m_1+m_2) \\ \vec{r}_2 = \vec{R}_{\text{COM}} + m_1\vec{r}/(m_1+m_2) \\ X_{\text{COM}} = (m_1x_1 + m_1x_2)/(m_1+m_2)\end{cases}$$

Calling the total mass $m_1 + m_2 = M$ and reduced mass $m_1m_2/(m_1+m_2) = \mu$, we have

$$\boxed{\mathcal{L} = \underbrace{\frac{1}{2}M\dot{\vec{R}}_{\text{COM}}^2 + gMX_{\text{COM}}}_{\text{COM}} + \underbrace{\frac{1}{2}\mu\dot{\vec{r}}^2 + \frac{Gm_1m_2}{r}}_{\text{relative}}}$$

The Lagrangian splits into two parts which describe the center-of-mass and relative dynamics, respectively. This makes sense physically because both bodies are essentially in "free fall" with each other. The center of mass of the system is decoupled from the relative motion, i.e. we can go to a frame in which the center of mass is stationary, and the only dynamics left is the relative motion of the masses.

(b) Going to the center of mass frame, we have the following Lagrangian

$$\mathcal{L}_r = \frac{1}{2}\mu\dot{\vec{r}}^2 + \frac{Gm_1m_2}{r}.$$

To avoid taking derivatives of the basis vectors in spherical coordinates, we may write $\dot{\vec{r}} = \dot{x}^2 + \dot{y}^2 + \dot{z}^2$ where $(x, y, z) = r(\sin\theta\cos\phi, \sin\theta\sin\phi, \cos\theta)$ and let Mathematica compute the more familiar derivatives. The result is

$$\boxed{\mathcal{L}_r = \frac{1}{2}\mu\left[\dot{r}^2 + r^2\left(\dot{\theta}^2 + \dot{\phi}^2\sin^2\theta\right)\right] + \frac{Gm_1m_2}{r}}$$

The corresponding Euler-Lagrange equations are

$$\begin{aligned}\frac{d}{dt}\left(\frac{\partial\mathcal{L}_r}{\partial\dot{r}}\right) = \frac{\partial\mathcal{L}_r}{\partial r} &\implies \mu\ddot{r} = -\frac{Gm_1m_2}{r^2} + \mu r\left(\dot{\theta}^2 + \dot{\phi}^2\sin^2\theta\right) \\ \frac{d}{dt}\left(\frac{\partial\mathcal{L}_r}{\partial\dot{\theta}}\right) = \frac{\partial\mathcal{L}_r}{\partial\theta} &\implies 2\dot{r}\dot{\theta} + r\ddot{\theta} = r\dot{\phi}^2\cos\theta\sin\theta \\ \frac{d}{dt}\left(\frac{\partial\mathcal{L}_r}{\partial\dot{\phi}}\right) = \frac{\partial\mathcal{L}_r}{\partial\phi} &\implies \mu r\sin\theta\left[2\left(\dot{r}\sin\theta + r\dot{\theta}\cos\theta\right)\dot{\phi} + r\ddot{\phi}\sin\theta\right] = 0\end{aligned}$$

(c) The Hamiltonian corresponding to this Lagrangian is obtained via the Legendre transform. To do this, we first find the canonical momenta p_r, p_θ, p_ϕ in Mathematica using $p_i = \partial\mathcal{L}_r/\partial\dot{q}_i$.

$$p_r = \mu\dot{r} \qquad p_\theta = \mu r^2\dot{\theta} \qquad p_\phi = \mu r^2 \sin^2\theta\dot{\phi}$$

The Hamiltonian is

$$\begin{aligned}\mathcal{H}_r &= \left(p_r\dot{r} + p_\theta\dot{\theta} + p_\phi\dot{\phi}\right) - \mathcal{L}_r \\ &= \boxed{\frac{1}{2}\mu\left[\dot{r}^2 + r^2\left(\dot{\theta}^2 + \dot{\phi}^2\sin^2\theta\right)\right] - \frac{Gm_1m_2}{r}}\end{aligned}$$

Alternatively, we can get this Hamiltonian (which is the total energy) by recognizing that the kinetic part of the Lagrangian is quadratic and the potential is not velocity dependent.

To find the Hamiltonian equations of motion, we first express $\mathcal{H}_r$ in terms of the canonical momenta:

$$\mathcal{H}_r = \frac{p_r^2}{2\mu} + \frac{p_\theta^2}{2\mu r^2} + \frac{p_\phi^2}{2\mu r^2\sin^2\theta} - \frac{Gm_1m_2}{r}$$

With this, we find

$$\dot{r} = \frac{\partial\mathcal{H}_r}{\partial p_r} = \frac{p_r}{\mu}; \qquad \dot{\theta} = \frac{\partial\mathcal{H}}{\partial p_\theta} = \frac{p_\theta}{\mu r^2}; \qquad \dot{\phi} = \frac{p_\phi}{\mu r^2\sin^2\theta}$$

and

$$\dot{p}_r = -\frac{\partial\mathcal{H}_r}{\partial r} = \frac{p_\theta^2}{\mu r^3} + \frac{p_\phi^2}{\mu r^3\sin^2\theta} - \frac{Gm_1m_2}{r^2}; \qquad \dot{p}_\theta = -\frac{\partial\mathcal{H}_r}{\partial\theta} = \frac{p_\phi^2\cos\theta}{\mu r^2\sin^3\theta}; \qquad \dot{p}_\phi = -\frac{\partial\mathcal{H}}{\partial\phi} = 0.$$

(d) Mathematica code:

```
(* Problem 1 *)

(* KE, PE, and Lagrangian *)

In[1]:= KE = (\[Mu]/2)*(D[x[t], t]^2 + D[y[t], t]^2 + D[z[t], t]^2);

In[2]:= PE = -G*m1*m2/r[t];

In[3]:= L = KE - PE

Out[3]= (G m1 m2)/r[t] +
1/2 \[Mu] (Derivative[1][x][t]^2 + Derivative[1][y][t]^2 +
Derivative[1][z][t]^2)

In[4]:= L =
L /. {x[t] -> r[t]*Sin[\[Theta][t]]*Cos[\[Phi][t]],
y[t] -> r[t]*Sin[\[Theta][t]]*Sin[\[Phi][t]],
z[t] -> r[t]*Cos[\[Theta][t]],
x'[t] -> D[r[t]*Sin[\[Theta][t]]*Cos[\[Phi][t]], t],
y'[t] -> D[r[t]*Sin[\[Theta][t]]*Sin[\[Phi][t]], t],
z'[t] -> D[r[t]*Cos[\[Theta][t]], t]} // FullSimplify

Out[4]= 1/2 ((2 G m1 m2)/
r[t] + \[Mu] Derivative[1][r][t]^2 + \[Mu] r[
t]^2 (Derivative[1][\[Theta]][t]^2 +
Sin[\[Theta][t]]^2 Derivative[1][\[Phi]][t]^2))

In[5]:= (* The 'r' equation *)

In[6]:= D[D[L, r'[t]], t] // FullSimplify

Out[6]= \[Mu] (r^\[Prime]\[Prime])[t]

In[7]:= D[L, r[t]] // FullSimplify
```

```
Out[7]= -((G m1 m2)/
r[t]^2) + \[Mu] r[
t] (Derivative[1][\[Theta]][t]^2 +
Sin[\[Theta][t]]^2 Derivative[1][\[Phi]][t]^2)

In[8]:= (* The 'Theta' equation *)

In[9]:= D[D[L, \[Theta]'[t]], t] // FullSimplify

Out[9]= \[Mu] r[
t] (2 Derivative[1][r][t] Derivative[1][\[Theta]][t] +
r[t] (\[Theta]^\[Prime]\[Prime])[t])

In[10]:= D[L, \[Theta][t]] // FullSimplify

Out[10]= \[Mu] Cos[\[Theta][t]] r[t]^2 Sin[\[Theta][t]] Derivative[
1][\[Phi]][t]^2

(* The 'Phi' equation *)

In[11]:= D[D[L, \[Phi]'[t]], t] // FullSimplify

Out[11]= \[Mu] r[
t] Sin[\[Theta][
t]] (2 (Sin[\[Theta][t]] Derivative[1][r][t] +
Cos[\[Theta][t]] r[t] Derivative[1][\[Theta]][t]) Derivative[
1][\[Phi]][t] +
r[t] Sin[\[Theta][t]] (\[Phi]^\[Prime]\[Prime])[t])

In[12]:= D[L, \[Phi][t]] // FullSimplify

Out[12]= 0

(* Canonical momenta *)

In[13]:= pr[t] = D[L, r'[t]]

Out[13]= \[Mu] Derivative[1][r][t]

In[14]:= p\[Theta][t] = D[L, \[Theta]'[t]]

Out[14]= \[Mu] r[t]^2 Derivative[1][\[Theta]][t]

In[15]:= p\[Phi][t] = D[L, \[Phi]'[t]]

Out[15]= \[Mu] r[t]^2 Sin[\[Theta][t]]^2 Derivative[1][\[Phi]][t]

(* Lagrangian to Hamiltonian *)

In[17]:= H = (pr[t]*r'[t] + p\[Theta][t]*\[Theta]'[t] +
p\[Phi][t]*\[Phi]'[t]) - L // Expand

Out[17]= -((G m1 m2)/r[t]) + 1/2 \[Mu] Derivative[1][r][t]^2 +
1/2 \[Mu] r[t]^2 Derivative[1][\[Theta]][t]^2 +
1/2 \[Mu] r[t]^2 Sin[\[Theta][t]]^2 Derivative[1][\[Phi]][t]^2

In[63]:= (* Velocities: new instances of P to put in Hamiltonian *)

In[18]:= velocities =
Solve[{Pr[t] == D[L, r'[t]], P\[Theta][t] == D[L, \[Theta]'[t]],
P\[Phi][t] == D[L, \[Phi]'[t]]}, {r'[t], \[Theta]'[t], \[Phi]'[
t]}][[1]]

Out[18]= {Derivative[1][r][t] -> Pr[t]/\[Mu],
Derivative[1][\[Theta]][t] -> P\[Theta][t]/(\[Mu] r[t]^2),
Derivative[1][\[Phi]][t] -> (
Csc[\[Theta][t]]^2 P\[Phi][t])/(\[Mu] r[t]^2)}

(* Write Hamiltonian in terms of momenta: *)

In[19]:= H = H /. velocities // Expand

Out[19]= Pr[t]^2/(2 \[Mu]) + P\[Theta][t]^2/(2 \[Mu] r[t]^2) + (
Csc[\[Theta][t]]^2 P\[Phi][t]^2)/(2 \[Mu] r[t]^2) - (G m1 m2)/r[t]

(*Hamiltonian EOMs*)

In[20]:= D[r[t], t] == D[H, Pr[t]]

Out[20]= Derivative[1][r][t] == Pr[t]/\[Mu]
```

```
In[21]:= D[\[Theta][t], t] == D[H, P\[Theta][t]]

Out[21]= Derivative[1][\[Theta]][t] == P\[Theta][t]/(\[Mu] r[t]^2)

In[22]:= D[\[Phi][t], t] == D[H, P\[Phi][t]]

Out[22]= Derivative[1][\[Phi]][t] == (
Csc[\[Theta][t]]^2 P\[Phi][t])/(\[Mu] r[t]^2)

In[26]:= D[Pr[t], t] == -D[H, r[t]] // Expand

Out[26]= Derivative[1][Pr][t] ==
P\[Theta][t]^2/(\[Mu] r[t]^3) + (
Csc[\[Theta][t]]^2 P\[Phi][t]^2)/(\[Mu] r[t]^3) - (G m1 m2)/r[t]^2

In[27]:= D[P\[Theta][t], t] == -D[H, \[Theta][t]] // Expand

Out[27]= Derivative[1][P\[Theta]][t] == (
Cot[\[Theta][t]] Csc[\[Theta][t]]^2 P\[Phi][t]^2)/(\[Mu] r[t]^2)

In[83]:= D[P\[Phi][t], t] == -D[H, \[Phi][t]] // Expand

Out[83]= Derivative[1][P\[Phi]][t] == 0
```

---

2. Double Pendulum in a Plane with Gravity

(a) In rectangular coordinates:

$$\begin{aligned}\mathcal{L} &= T - U \\ &= \frac{1}{2}m_1(\dot{x}_1^2 + \dot{y}_1^2) + \frac{1}{2}m_2(\dot{x}_2^2 + \dot{y}_2^2) - m_1 g y_1 - m_2 g y_2.\end{aligned}$$

With

$$\begin{aligned} x_1 &= l_1 \sin\theta_1; & x_2 &= l_2 \sin\theta_2 + l_1 \sin\theta_1 \\ y_1 &= -l_1 \cos\theta_1; & y_2 &= -l_2\cos\theta_2 - l_1\cos\theta_1 \end{aligned}$$

we have

$$\boxed{\mathcal{L} = \frac{1}{2}(m_1+m_2)l_1^2\dot{\theta}_1^2 + \frac{1}{2}l_1^2 m_2\dot{\theta}_2^2 + l_1 l_2 m_2\cos(\theta_1-\theta_2)\dot{\theta}_1\dot{\theta}_2 + g l_1(m_1+m_2)\cos\theta_1 + g l_2 m_2\cos\theta_2}$$

The equations of motion are:

$$\frac{d}{dt}\left(\frac{\partial\mathcal{L}}{\partial\dot{\theta}_1}\right) = \frac{\partial\mathcal{L}}{\partial\theta_1} \implies l_2 m_2\sin(\theta_1-\theta_2)\dot{\theta}_2^2 + (m_1+m_2)(g\sin\theta_1 + l_1\ddot{\theta}_1) + l_2 m_2\cos(\theta_1-\theta_2)\ddot{\theta}_2 = 0$$

$$\frac{d}{dt}\left(\frac{\partial\mathcal{L}}{\partial\dot{\theta}_2}\right) = \frac{\partial\mathcal{L}}{\partial\theta_1} \implies g\sin\theta_2 - l_1\sin(\theta_1-\theta_2)\dot{\theta}_1^2 + l_1\cos(\theta_1-\theta_2)\ddot{\theta}_1 + l_2\ddot{\theta}_2 = 0.$$

(b) Now take $m_1 = m_2 = m$. Following a similar procedure as before, we first find the canonical momenta using $p_i = \partial\mathcal{L}/\partial\dot{q}_i$, then find the Hamiltonian by Legendre-transforming the Lagrangian. Equivalently, we can simply take the total energy, as the kinetic part of the Lagrangian is quadratic and the potential is velocity independent.

$$\boxed{\mathcal{H} = l_1^2 m\dot{\theta}_1^2 + \frac{1}{2}l_2^2 m\dot{\theta}_2^2 + l_1 l_2 m\cos(\theta_1-\theta_2)\dot{\theta}_1\dot{\theta}_2 - 2g l_1 m\cos\theta_1 - g l_2 m\cos\theta_2}$$

To write $\mathcal{H}$ in terms of the canonical momenta, we need to find how they are related to the velocities. Setting $p_{\theta_i} = \partial\mathcal{L}/\partial\dot{\theta}_i$ we find

$$\dot{\theta}_1 = \frac{-l_2 p_{\theta_1} + l_1\cos(\theta_1-\theta_2)p_{\theta_2}}{l_1^2 l_2 m[\cos^2(\theta_1-\theta_2)-2]} \qquad \dot{\theta}_2 = \frac{l_2\cos(\theta_1-\theta_2)p_{\theta_1} - 2l_1 p_{\theta_2}}{l_1 l_2^2 m[\cos^2(\theta_1-\theta_2)-2]}$$

In terms of p_{θ_1} and p_{θ_2}, the Hamiltonian is

$$\boxed{\mathcal{H} = -\frac{l_1^2\left(g{l_2}^2m^2[\cos(2(\theta_1-\theta_2))-3](2l_1\cos\theta_1+l_2\cos\theta_2)+2p_{\theta_2}^2\right)-2l_1l_2p_{\theta_1}p_{\theta_2}\cos(\theta_1-\theta_2)+l_2^2p_{\theta_1}^2}{l_1^2l_2^2m[\cos(2(\theta_1-\theta_2))-3]}}$$

The remain equations of motion are the "$\dot{p}$" equations:

$$\dot{p}_{\theta_1} = -\frac{\partial\mathcal{H}}{\partial\theta_1} = \frac{-2gl_1^3l_2^2m^2\sin\theta_1[\cos(2(\theta_1-\theta_2))-3]^2+2\sin(2(\theta_1-\theta_2))\left(2l_1^2p_{\theta_2}^2+l_2^2p_{\theta_1}^2\right)}{l_1^2l_2^2m[\cos(2(\theta_1-\theta_2))-3]^2} + \frac{-2l_1l_2p_{\theta_1}p_{\theta_2}\sin(\theta_1-\theta_2)[\cos(2(\theta_1-\theta_2))+5]}{l_1^2l_2^2m[\cos(2(\theta_1-\theta_2))-3]^2}$$

and

$$\dot{p}_{\theta_2} = -\frac{\partial\mathcal{H}}{\partial\theta_2} = -gl_2m\sin\theta_2 + \frac{2\sin(\theta_1-\theta_2)\left(-4l_1^2p_{\theta_2}^2\cos(\theta_1-\theta_2)+l_1l_2p_{\theta_1}p_{\theta_2}[\cos(2(\theta_1-\theta_2))+5]-2l_2^2p_{\theta_1}^2\cos(\theta_1-\theta_2)\right)}{l_1^2l_2^2m[\cos(2(\theta_1-\theta_2))-3]^2}$$

(c) Mathematica code:

```
(* Problem 2 *)

(* KE, PE, and Lagrangian *)

In[1]:= KE = (1/2)*m1*(D[x1[t], t]^2 + D[y1[t], t]^2) + (1/2)*
m2*(D[x2[t], t]^2 + D[y2[t], t]^2);

In[2]:= PE = m1*g*y1[t] + m2*g*y2[t];

In[3]:= L = KE - PE;

In[4]:= L = L /. {
x1[t] -> l1*Sin[\[Theta]1[t]],
y1[t] -> -l1*Cos[\[Theta]1[t]],
x2[t] -> l2*Sin[\[Theta]2[t]] + l1*Sin[\[Theta]1[t]],
y2[t] -> -l2*Cos[\[Theta]2[t]] - l1*Cos[\[Theta]1[t]],
x1'[t] -> D[l1*Sin[\[Theta]1[t]], t],
y1'[t] -> D[-l1*Cos[\[Theta]1[t]], t],
x2'[t] -> D[l2*Sin[\[Theta]2[t]] + l1*Sin[\[Theta]1[t]], t],
y2'[t] -> D[-l2*Cos[\[Theta]2[t]] - l1*Cos[\[Theta]1[t]], t]} //
FullSimplify

Out[4]= 1/2 (2 g (l1 (m1 + m2) Cos[\[Theta]1[t]] +
l2 m2 Cos[\[Theta]2[t]]) +
l1^2 (m1 + m2) Derivative[1][\[Theta]1][t]^2 +
2 l1 l2 m2 Cos[\[Theta]1[t] - \[Theta]2[t]] Derivative[
1][\[Theta]1][t] Derivative[1][\[Theta]2][t] +
l2^2 m2 Derivative[1][\[Theta]2][t]^2)

(* Lagrangian EOMs *)

In[5]:= D[D[L, \[Theta]1'[t]], t] == D[L, \[Theta]1[t]] // FullSimplify

Out[5]= l1 (l2 m2 Sin[\[Theta]1[t] - \[Theta]2[t]] Derivative[
1][\[Theta]2][
t]^2 + (m1 + m2) (g Sin[\[Theta]1[t]] +
l1 (\[Theta]1^\[Prime]\[Prime])[t]) +
l2 m2 Cos[\[Theta]1[t] - \[Theta]2[
t]] (\[Theta]2^\[Prime]\[Prime])[t]) == 0

In[6]:= D[D[L, \[Theta]2'[t]], t] == D[L, \[Theta]2[t]] // FullSimplify

Out[6]= l2 m2 (g Sin[\[Theta]2[t]] -
l1 Sin[\[Theta]1[t] - \[Theta]2[t]] Derivative[1][\[Theta]1][
t]^2 + l1 Cos[\[Theta]1[t] - \[Theta]2[
t]] (\[Theta]1^\[Prime]\[Prime])[t] +
```

```
l2 (\[Theta]2^\[Prime]\[Prime])[t]) == 0

(* Take m1 = m2 = m *)

In[7]:= D[D[L, \[Theta]1'[t]], t] == D[L, \[Theta]1[t]] /. {m1 -> m,
m2 -> m} // FullSimplify

Out[7]= l1 m (2 g Sin[\[Theta]1[t]] +
l2 Sin[\[Theta]1[t] - \[Theta]2[t]] Derivative[1][\[Theta]2][
t]^2 + 2 l1 (\[Theta]1^\[Prime]\[Prime])[t] +
l2 Cos[\[Theta]1[t] - \[Theta]2[t]] (\[Theta]2^\[Prime]\[Prime])[
t]) == 0

In[8]:= D[D[L, \[Theta]2'[t]], t] == D[L, \[Theta]2[t]] /. {m1 -> m,
m2 -> m} // FullSimplify

Out[8]= l2 m (g Sin[\[Theta]2[t]] -
l1 Sin[\[Theta]1[t] - \[Theta]2[t]] Derivative[1][\[Theta]1][
t]^2 + l1 Cos[\[Theta]1[t] - \[Theta]2[
t]] (\[Theta]1^\[Prime]\[Prime])[t] +
l2 (\[Theta]2^\[Prime]\[Prime])[t]) == 0

(*Hamiltonian*)

In[9]:= H = (D[L, \[Theta]1'[t]]*D[\[Theta]1[t], t] +
D[L, \[Theta]2'[t]]*D[\[Theta]2[t], t]) - L /. {m1 -> m,
m2 -> m} // Expand

Out[9]= -2 g l1 m Cos[\[Theta]1[t]] - g l2 m Cos[\[Theta]2[t]] +
l1^2 m Derivative[1][\[Theta]1][t]^2 +
l1 l2 m Cos[\[Theta]1[t] - \[Theta]2[t]] Derivative[1][\[Theta]1][
t] Derivative[1][\[Theta]2][t] +
1/2 l2^2 m Derivative[1][\[Theta]2][t]^2

(*solve for velocities in terms of momenta to write H in terms of \
momenta*)

In[10]:= D[L, \[Theta]1'[t]] /. {m1 -> m, m2 -> m} // FullSimplify

Out[10]= l1 m (2 l1 Derivative[1][\[Theta]1][t] +
l2 Cos[\[Theta]1[t] - \[Theta]2[t]] Derivative[1][\[Theta]2][t])

In[11]:= velocities =
Solve[{P\[Theta]1[t] == D[L, \[Theta]1'[t]],
P\[Theta]2[t] == D[L, \[Theta]2'[t]]}, {\[Theta]1'[
t], \[Theta]2'[t]}][[1]] /. {m1 -> m, m2 -> m} // FullSimplify

Out[11]= {Derivative[1][\[Theta]1][t] -> (-l2 P\[Theta]1[t] +
l1 Cos[\[Theta]1[t] - \[Theta]2[t]] P\[Theta]2[t])/(
l1^2 l2 m (-2 + Cos[\[Theta]1[t] - \[Theta]2[t]]^2)),
Derivative[1][\[Theta]2][t] -> (
l2 Cos[\[Theta]1[t] - \[Theta]2[t]] P\[Theta]1[t] -
2 l1 P\[Theta]2[t])/(
l1 l2^2 m (-2 + Cos[\[Theta]1[t] - \[Theta]2[t]]^2))}

In[12]:= H = H /. velocities // FullSimplify

Out[12]= -((l2^2 P\[Theta]1[t]^2 -
2 l1 l2 Cos[\[Theta]1[t] - \[Theta]2[t]] P\[Theta]1[
t] P\[Theta]2[t] +
l1^2 (g l2^2 m^2 (-3 +
Cos[2 (\[Theta]1[t] - \[Theta]2[t])]) (2 l1 Cos[\[Theta]1[
t]] + l2 Cos[\[Theta]2[t]]) +
2 P\[Theta]2[t]^2))/(l1^2 l2^2 m (-3 +
Cos[2 (\[Theta]1[t] - \[Theta]2[t])])))

(* Hamiltonian EOMs *)

In[16]:= Solve[P\[Theta]1'[t] == -D[H, \[Theta]1[t]],
P\[Theta]1'[t]][[1]] // FullSimplify

Out[16]= {Derivative[1][P\[Theta]1][
t] -> (-2 g l1^3 l2^2 m^2 (-3 +
Cos[2 (\[Theta]1[t] - \[Theta]2[t])])^2 Sin[\[Theta]1[t]] -
2 l1 l2 (5 + Cos[2 (\[Theta]1[t] - \[Theta]2[t])]) P\[Theta]1[
t] P\[Theta]2[t] Sin[\[Theta]1[t] - \[Theta]2[t]] +
2 (l2^2 P\[Theta]1[t]^2 + 2 l1^2 P\[Theta]2[t]^2) Sin[
2 (\[Theta]1[t] - \[Theta]2[t])])/(l1^2 l2^2 m (-3 +
Cos[2 (\[Theta]1[t] - \[Theta]2[t])])^2)}

In[17]:= Solve[P\[Theta]2'[t] == -D[H, \[Theta]2[t]],
```

```
P\[Theta]2'[t]][[1]] // FullSimplify

Out[17]= {Derivative[1][P\[Theta]2][
t] -> (2 (-2 l2^2 Cos[\[Theta]1[t] - \[Theta]2[t]] P\[Theta]1[
t]^2 + l1 l2 (5 +
Cos[2 (\[Theta]1[t] - \[Theta]2[t])]) P\[Theta]1[
t] P\[Theta]2[t] -
4 l1^2 Cos[\[Theta]1[t] - \[Theta]2[t]] P\[Theta]2[
t]^2) Sin[\[Theta]1[t] - \[Theta]2[t]])/(l1^2 l2^2 m (-3 +
Cos[2 (\[Theta]1[t] - \[Theta]2[t])])^2) -
g l2 m Sin[\[Theta]2[t]]}

In[66]:= \[Theta]1'[t] == D[H, P\[Theta]1[t]] // FullSimplify

Out[66]= l1 Derivative[1][\[Theta]1][t] == (
2 l2 P\[Theta]1[t] -
2 l1 Cos[\[Theta]1[t] - \[Theta]2[t]] P\[Theta]2[t])/(
3 l1 l2 m - l1 l2 m Cos[2 \[Theta]1[t] - 2 \[Theta]2[t]])

In[67]:= \[Theta]2'[t] == D[H, P\[Theta]2[t]] // FullSimplify

Out[67]= Derivative[1][\[Theta]2][t] == (
2 (l2 Cos[\[Theta]1[t] - \[Theta]2[t]] P\[Theta]1[t] -
2 l1 P\[Theta]2[t]))/(
l1 l2^2 m (-3 + Cos[2 (\[Theta]1[t] - \[Theta]2[t])]))
```

---

3. Point Mass on a Hoop: Goldstein Ch.2 Problem #18.
By the geometry of the problem, we the system may be described by one (spherical) coordinate θ defined as usual. r is fixed at $r = a$ and $\phi(t) = \omega t$, where ω is fixed.

$$(x, y, z) = (a \sin\theta \cos\omega t, a \sin\theta \sin\omega t, a \cos\theta).$$

With this, the Lagrangian is

$$\begin{aligned}\mathcal{L} &= \frac{1}{2} m \left[\dot{x}^2 + \dot{y}^2 + \dot{z}^2\right] - mgz \\ &= \frac{1}{2} am \left[-2g\cos\theta + a\omega^2 \sin^2\theta + a\dot{\theta}^2\right].\end{aligned}$$

There is only one Lagrangian equation of motion:

$$\frac{d}{dt}\left(\frac{\partial \mathcal{L}}{\partial \dot{\theta}}\right) = \frac{\partial \mathcal{L}}{\partial \theta} \implies a\ddot{\theta} = \left(g + a\omega^2 \cos\theta\right)\sin\theta.$$

It is clear from the functional form of $\mathcal{L}$ that a constant of motion is r since $r = a$ fixed, and that $\mathcal{H} = T + V =$ total energy. Moreover, since $\mathcal{L}$ is time-independent, we have conservation of energy, making energy a constant of motion. We thus have

$$\text{const} = \mathcal{H} = \frac{\partial \mathcal{L}}{\partial \dot{\theta}}\dot{\theta} - \mathcal{L} = \frac{1}{2} am \left(a\dot{\theta}^2 - a\omega^2 \sin^2\theta + 2g\cos\theta\right).$$

We now want to find the critical value ω_0 described in the problem statement. Since the particle is stationary, $\dot{\theta} = 0$. The particle is at a stationary point exactly when it is at a local minimum of some "effective" potential $V(\theta)$. From the Hamiltonian $\mathcal{H} = \text{KE} + V(\theta)$, we can read off this effective potential:

$$V(\theta) = \frac{1}{2} am \left(-a\omega^2 \sin^2\theta + 2g\cos\theta\right).$$

Since

$$\frac{\partial V(\theta)}{\partial \theta} = \frac{1}{2} am \left(-a\omega^2 \sin 2\theta - 2g\sin\theta\right) = am \sin\theta \left(-a\omega^2 \cos\theta - g\right),$$

the stationary points are $\theta = 0, \pi$ or $\theta = \arccos(-\omega_0^2/\omega^2)$ where $\boxed{\omega_0 \equiv \sqrt{g/a}}$. Of these three points, $\theta = 0$ is always unstable because for $\theta \approx 0$, $V(\theta)$ looks like $\cos\theta$ whose first derivative near $\theta = 0$ is negative, so the particle will move away from $\theta = 0$.

When $\omega \le \omega_0$, the only equilibrium is $\theta = \pi$ since $\partial V(\theta)/\partial\theta \le 0$ for all $\theta \in [0, \pi]$. When $\omega > \omega_0$, the equilibrium point becomes $\theta = \theta_0 = \arccos(-\omega_0^2/\omega^2)$ because $\partial V/\partial\theta \ge 0$ for $\theta \ge \theta_0$.

Mathematica code:

```
(*Problem 3*)
In[20]:= KE = (m/2)*(D[x[t], t]^2 + D[y[t], t]^2 + D[z[t], t]^2);
In[21]:= PE = m*g*z[t];
In[22]:= L = KE - PE
Out[22]= -g m z[t] +
1/2 m (Derivative[1][x][t]^2 + Derivative[1][y][t]^2 +
Derivative[1][z][t]^2)
(*Lagrangian*)
In[23]:= L = L /. {x[t] -> a*Sin[\[Theta][t]]*Cos[\[Omega]*t],
y[t] -> a*Sin[\[Theta][t]]*Sin[\[Omega]*t],
z[t] -> a*Cos[\[Theta][t]],
x'[t] -> D[a*Sin[\[Theta][t]]*Cos[\[Omega]*t], t],
y'[t] -> D[a*Sin[\[Theta][t]]*Sin[\[Omega]*t], t],
z'[t] -> D[a*Cos[\[Theta][t]], t]} // FullSimplify
Out[23]= 1/2 a m (-2 g Cos[\[Theta][t]] + a \[Omega]^2 Sin[\[Theta][t]]^2 +
a Derivative[1][\[Theta]][t]^2)
In[19]:= (*Hamiltonian*)
In[25]:= H = D[L, \[Theta]'[t]]*\[Theta]'[t] - L // FullSimplify
Out[25]= 1/2 a m (2 g Cos[\[Theta][t]] - a \[Omega]^2 Sin[\[Theta][t]]^2 +
a Derivative[1][\[Theta]][t]^2)
In[26]:= 1/2 a m (2 g Cos[\[Theta][t]] - a \[Omega]^2 Sin[\[Theta][t]]^2 +
a Derivative[1][\[Theta]][t]^2) // TeXForm
Out[26]//TeXForm=
\frac{1}{2} a m \left(a \theta '(t)^2-a \omega ^2 \sin ^2(\theta
(t))+2 g \cos (\theta (t))\right)
(*Lagrangian EOM*)
In[88]:= FullSimplify[D[D[L, \[Theta]'[t]], t] == D[L, \[Theta][t]]]
Out[88]= a^2 m (\[Theta]^\[Prime]\[Prime])[t] ==
a m (g + a \[Omega]^2 Cos[\[Theta][t]]) Sin[\[Theta][t]]
```

---

4. Spring System on a Plane

(a) The Lagrangian in Cartesian coordinates is

$$\boxed{\mathcal{L} = \frac{1}{2}m_1(\dot{x}_1^2 + \dot{y}_1^2) + \frac{1}{2}m_2(\dot{x}_2^2 + \dot{y}_2^2) - \frac{1}{2}k\left(\sqrt{(x_1 - x_2)^2 + (y_1 - y_2)^2} - b\right)^2}$$

(b) From previous problems, we know that the Lagrangian can be written as

$$\mathcal{L} = -\frac{k}{2}(b - r)^2 + \frac{1}{2}\frac{m_1 m_2}{m_1 + m_2}\dot{\vec{r}}^2 + \frac{1}{2}(m_1 + m_2)\dot{\vec{R}}_{\mathrm{COM}}^2$$

where $\vec{R}_{\mathrm{COM}}$ denotes the center of mass position vector. This motivates us to pick the following coordinates: $(x_{\mathrm{COM}}, y_{\mathrm{COM}}, r, \theta)$, where $(x_{\mathrm{COM}}, y_{\mathrm{COM}})$ describes the position of the center of mass of the system, while (r, θ) together describe the relative position vector of the two masses. These new coordinates are defined by

$$x_{\mathrm{COM}} = \frac{m_1 x_1 + m_2 x_2}{m_1 + m_2}; \quad y_{\mathrm{COM}} = \frac{m_1 y_1 + m_2 y_2}{m_1 + m_2}; \quad r = \sqrt{(x_1 - x_2)^2 + (y_1 - y_2)^2}; \quad \theta = \arctan\frac{y_2 - y_1}{x_2 - x_1}.$$

As a result,

$$\boxed{\mathcal{L} = -\frac{k}{2}(b - r)^2 + \frac{1}{2}\frac{m_1 m_2}{m_1 + m_2}\left(\dot{r}^2 + r^2\dot{\theta}^2\right) + \frac{1}{2}(m_1 + m_2)\left(\dot{x}_{\mathrm{COM}}^2 + \dot{y}_{\mathrm{COM}}^2\right)}$$

The equations of motion are:

$$\frac{d}{dt}\frac{\partial \mathcal{L}}{\partial \dot{x}_{\text{COM}}} = \frac{\partial \mathcal{L}}{\partial x_{\text{COM}}} \implies \ddot{x}_{\text{COM}} = 0$$

$$\frac{d}{dt}\frac{\partial \mathcal{L}}{\partial \dot{y}_{\text{COM}}} = \frac{\partial \mathcal{L}}{\partial y_{\text{COM}}} \implies \ddot{y}_{\text{COM}} = 0$$

$$\frac{d}{dt}\frac{\partial \mathcal{L}}{\partial \dot{r}} = \frac{\partial \mathcal{L}}{\partial r} \implies \ddot{r} = \frac{k(m_1+m_2)(b-r)}{m_1 m_2} + r\dot{\theta}^2$$

$$\frac{d}{dt}\frac{\partial \mathcal{L}}{\partial \dot{\theta}} = \frac{\partial \mathcal{L}}{\partial \theta} \implies \ddot{\theta} = -\frac{2\dot{r}\dot{\theta}}{r}$$

(c) There are $\boxed{3}$ cyclic coordinates: x_{COM}, y_{COM}, and θ, since $\mathcal{L}$ does not explicitly depend on them. The **three conserved generalized momenta** are thus

$$p_{x_{\text{COM}}} = \frac{\partial \mathcal{L}}{\partial \dot{x}_{\text{COM}}} = (m_1+m_2)\dot{x}_{\text{COM}}; \qquad p_{y_{\text{COM}}} = \frac{\partial \mathcal{L}}{\partial \dot{y}_{\text{COM}}} = (m_1+m_2)\dot{y}_{\text{COM}}; \qquad p_\theta = \frac{\partial \mathcal{L}}{\partial \dot{\theta}} = \frac{m_1 m_2}{m_1+m_2} r^2\dot{\theta}.$$

We hope to show that there is a solution which rotates but does not oscillate. To this end, we will use energy conservation as a constraint on the possible dynamics of the system. In particular, since we are interested in how $r(t)$, we will look at the r-equation and the force which dictates r's behavior. Letting $\mu = m_1 m_2/(m_1+m_2)$,

$$\mu\ddot{r} = k(b-r) + r\dot{\theta}^2.$$

Consider this a one-body problem with Newton's second law $\vec{F} = m\vec{a}$. This force is associated with the potential $V(r)$ where

$$V(r) = \frac{1}{2}k(b-r)^2 - \frac{1}{2}\mu r^2\dot{\theta}^2.$$

On the other hand, we may compute the total energy of the system. After recognizing that the Lagrangian is quadratic in the kinetic terms and has velocity independent potentials, the total energy is the Hamiltonian:

$$\mathcal{H} = \frac{1}{2}\mu\dot{r}^2 - \frac{1}{2}\mu r^2\dot{\theta}^2 + \frac{1}{2}k(b-r)^2 + \text{some constant (COM)}$$

where we have used the fact that $\ddot{x}_{\text{COM}} = \ddot{y}_{\text{COM}} = 0$. Since $\mathcal{H}$ has no explicit time dependence, $d\mathcal{H}/dt = 0$, and thus energy is conserved. In particular, when we compare $\mathcal{H}$ to $V(r)$, we find that

$$\frac{1}{2}\mu\dot{r}^2 + V(r) = \text{some (other) constant}$$

Now, suppose $\dot{r}(0) = 0$ (does not oscillate) but $\dot{\theta} \neq 0$ (rotates). If $k > \mu\dot{\theta}^2$ then $\partial^2 V/\partial r > 0$, ensuring a stable equilibrium. Thus, if we start at $r(0) = r_0 = bk/(k-\mu\dot{\theta}^2)$ where $\partial V/\partial r|_{r_0} = 0$ then $\dot{r}$ remains at zero for all $t > 0$. Therefore, there exists a solution which rotates but does not oscillate.

When $\dot{\theta}$ gets large, i.e., when $\dot{\theta} \geq \sqrt{k/\mu}$, then the concavity of $V(r)$ changes to nonpositive and the stable equilibrium no longer exists (when $b = 0$ and $\dot{\theta} = \sqrt{k/\mu}$ then the potential becomes flat with no tilt). r diverges to infinity in this case. Intuitively this means that the rotation is so fast that the spring can no longer counteract the centrifugal force. Alternatively, one could sketch out the graph of $V(\theta)$ for various values of θ and make similar conclusions (see the end of the Mathematica code).

(d) Mathematica code:

```
(*Problem 4*)

In[50]:= KE = (1/2)*(m1 + m2)*(D[xCOM[t], t]^2 +
D[yCOM[t], t]^2) + (1/2)*(m1*
m2)*(D[r[t], t]^2 + r[t]^2*D[\[Theta][t], t]^2)/(m1 + m2);

In[51]:= PE = (k/2)*(r[t] - b)^2;

In[52]:= L = KE - PE;

(*Lagrangian*)

In[53]:= L // FullSimplify

Out[53]= 1/2 (-k (b - r[t])^2 + (m1 + m2) (Derivative[1][xCOM][t]^2 +
Derivative[1][yCOM][t]^2) + (
m1 m2 (Derivative[1][r][t]^2 +
r[t]^2 Derivative[1][\[Theta]][t]^2))/(m1 + m2))

(*Lagrangian EOMs*)

(*xCOM equation*)

In[54]:= Solve[FullSimplify[D[D[L, xCOM'[t]], t] == D[L, xCOM[t]]],
xCOM''[t]][[1]] // Expand

Out[54]= {(xCOM^\[Prime]\[Prime])[t] -> 0}

(*yCOM equation*)

In[55]:= Solve[FullSimplify[D[D[L, yCOM'[t]], t] == D[L, yCOM[t]]],
yCOM''[t]][[1]] // Expand

Out[55]= {(yCOM^\[Prime]\[Prime])[t] -> 0}

(*r equation*)

In[65]:= Solve[FullSimplify[D[D[L, r'[t]], t] == D[L, r[t]]],
r''[t]][[1]] // FullSimplify

Out[65]= {(r^\[Prime]\[Prime])[t] -> (k (m1 + m2) (b - r[t]))/(
m1 m2) + r[t] Derivative[1][\[Theta]][t]^2}

In[15]:= (*theta equation*)

In[57]:= Solve[
FullSimplify[
D[D[L, \[Theta]'[t]], t] == D[L, \[Theta][t]]], \[Theta]''[t]][[
1]] // Expand

Out[57]= {(\[Theta]^\[Prime]\[Prime])[t] -> -((
2 Derivative[1][r][t] Derivative[1][\[Theta]][t])/r[t])}

(*Cyclic coordinates*)

In[58]:= D[L, xCOM[t]]

Out[58]= 0

In[59]:= D[L, yCOM[t]]

Out[59]= 0

In[60]:= D[L, \[Theta][t]]

Out[60]= 0

(*Conserved generalized momenta*)

In[61]:= D[L, xCOM'[t]] // Simplify

Out[61]= (m1 + m2) Derivative[1][xCOM][t]

In[62]:= D[L, yCOM'[t]] // Simplify

Out[62]= (m1 + m2) Derivative[1][yCOM][t]

In[63]:= D[L, \[Theta]'[t]] // Simplify

Out[63]= (m1 m2 r[t]^2 Derivative[1][\[Theta]][t])/(m1 + m2)
```

```
In[1]:= (*minimum potential*)

In[1]:= V[r_] = (1/2)*k*(b - r)^2 - (1/2)*\[Mu]*r^2*\[Theta]'^2;

In[2]:= D[V[r], r]

Out[2]= -k (b - r) - r \[Mu] (Derivative[1][\[Theta]])^2

In[3]:= Solve[D[V[r], r] == 0, r]

Out[3]= {{r -> (b k)/(k - \[Mu] (Derivative[1][\[Theta]])^2)}}

(*changing potential*)

k = 1; b = 2; \[Mu] = 1; \[Theta]' = Sqrt[k/\[Mu]];

Plot[(1/2)*k*(b - r)^2 - (1/2)*\[Mu]*r^2*\[Theta]'^2, {r, 0, 200}]
```

---

5. For 8.09 ONLY

---

6. Routhian Mechanics

We start with the definition of a particular Routhian:

$$R(q_1,\ldots,q_n,p_1,\ldots,p_s,\dot{q}_{s+1},\ldots,\dot{q}_n,t)=\sum_{k=1}^{s}p_k\dot{q}_k-\mathcal{L}(q_1,\ldots,q_n,\dot{q}_1,\ldots\dot{q}_n,t).$$

For $i=1,\ldots,n$ we have

$$\boxed{\frac{\partial R}{\partial q_i}}=-\frac{\partial\mathcal{L}}{\partial q_i}=-\frac{d}{dt}\left(\frac{\partial\mathcal{L}}{\partial\dot{q}_i}\right)=-\frac{d}{dt}p_i=\boxed{-\dot{p}_i}$$

For $i=1,\ldots,s$, we have

$$\boxed{\frac{\partial R}{\partial p_i}=\dot{q}_i}$$

For $i=s+1,\ldots,n$, we have

$$\boxed{\frac{\partial R}{\partial\dot{q}_i}}=-\frac{\partial\mathcal{L}}{\partial\dot{q}_i}=\boxed{-p_i}\implies\boxed{\frac{d}{dt}\frac{\partial R}{\partial\dot{q}_i}}=-\frac{d}{dt}\left(\frac{\partial\mathcal{L}}{\partial\dot{q}_i}\right)=-\frac{d}{dt}p_i=\boxed{\frac{\partial R}{\partial q_i}}$$

Finally,

$$\boxed{\frac{\partial R}{\partial t}=-\frac{\partial\mathcal{L}}{\partial t}}$$

---

7. Extra Problem: Equivalent Lagrangians

Suppose that $\mathcal{L}$ satisfies the Euler-Lagrange equations and F is a differentiable function. We claim that

$$\mathcal{L}'=\mathcal{L}+\frac{dF(q,t)}{dt}$$

also satisfies the Euler-Lagrange equations.

Proof. On the one hand,

$$\frac{d}{dt}\frac{\partial\mathcal{L}'}{\partial\dot{q}}=\frac{d}{dt}\frac{\partial\mathcal{L}}{\partial\dot{q}}+\frac{d}{dt}\frac{\partial}{\partial\dot{q}}\frac{dF}{dt}=\frac{d}{dt}\frac{\partial\mathcal{L}}{\partial\dot{q}}+\frac{d}{dt}\frac{\partial F}{\partial q}=\frac{d}{dt}\frac{\partial\mathcal{L}}{\partial\dot{q}}+\frac{\partial}{\partial q}\left(\frac{\partial F}{\partial q}\right)\dot{q}+\frac{\partial^2F}{\partial t\partial q},$$

where for the second equality we have used the fact that F does not depend on $\dot{q}$ (i.e., dF/dt generates a factor of $\dot{q}$, which gets taken away by $\partial/\partial\dot{q}$).

On the other hand,

$$\frac{\partial \mathcal{L}'}{\partial q} = \frac{\partial \mathcal{L}}{\partial q} + \frac{\partial}{\partial q}\frac{dF}{dt} = \frac{\partial \mathcal{L}}{\partial q} + \frac{\partial}{\partial q}\left(\frac{\partial F}{\partial q}\right)\dot{q} + \frac{\partial^2 F}{\partial q \partial t}.$$

Since $\partial^2 F/\partial q \partial t = \partial^2 F/\partial t \partial q$ and that $\mathcal{L}$ satisfies the Euler-Lagrange equations, we see that

$$\frac{d}{dt}\frac{\partial \mathcal{L}'}{\partial \dot{q}} = \frac{\partial \mathcal{L}'}{\partial q}$$

after comparing the two results. □